

Analysis of the Given Grammar

(February 20, 2022)

COLORS in this document

Black: for original rules in the grammar in Language Specification document

Red: specifies needs for modifications

Green: rules do not require modification

Brown: Not valid syntactically

The nonterminal $\langle \text{program} \rangle$ is the start symbol of the given grammar.

1. $\langle \text{program} \rangle \Rightarrow \langle \text{otherFunctions} \rangle \langle \text{mainFunction} \rangle$

The LL(1) property for the nonterminal $\langle \text{program} \rangle$ is satisfied trivially due to single production for it. The following FOLLOW set will be required for verifying LL(1) compatibility for the nonterminal $\langle \text{otherFunctions} \rangle$ in (3).

$\text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \text{FIRST}(\langle \text{mainFunction} \rangle) = \{\text{TK_MAIN}\}$

Note that all function definitions precede the main function definition and the language does not have constructs for function prototype declarations.

2. $\langle \text{mainFunction} \rangle \Rightarrow \text{TK_MAIN} \langle \text{stmts} \rangle \text{TK_END}$

The LL(1) property for the nonterminal $\langle \text{mainFunction} \rangle$ is satisfied trivially due to single production for it.

$\text{FIRST}(\langle \text{mainFunction} \rangle) = \text{FIRST}(\alpha) = \{\text{TK_MAIN}\}$

where α represents the right hand side of the production i.e. $\text{TK_MAIN} \langle \text{stmts} \rangle \text{TK_END}$.

3. $\langle \text{otherFunctions} \rangle \Rightarrow \langle \text{function} \rangle \langle \text{otherFunctions} \rangle \mid \text{eps}$

The nonterminal $\langle \text{otherFunctions} \rangle$ need special care for verifying the LL(1) compatibility due to a nullable production. Let us first verify that the sets $\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle)$ and $\text{FOLLOW}(\langle \text{otherFunctions} \rangle)$ are disjoint.

$\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle) = \text{FIRST}(\langle \text{function} \rangle) = \{\text{TK_FUNID}\}$

Note that $\langle \text{function} \rangle$ has no nullable production.

Also from 1, we have $\text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \{\text{TK_MAIN}\}$

i.e.

$$\text{FIRST}(\langle \text{function} \rangle \langle \text{otherFunctions} \rangle) \cap \text{FOLLOW}(\langle \text{otherFunctions} \rangle) = \phi$$

The given productions for $\langle \text{otherFunctions} \rangle$ are LL(1) compatible.

NOTE: Other properties such as ambiguity, left recursiveness, left factoring needs etc. causing violation of LL(1) compatibility of the rules will be discussed only if one or more of them exist. Otherwise I am focussing on violations due to epsilon productions. Also I will highlight the introduction of new nonterminals to incorporate the precedence of arithmetic operators and handling operations on variables of type record and union.

4. $\langle \text{function} \rangle ::= \text{TK_FUNID } \langle \text{input_par} \rangle \langle \text{output_par} \rangle \text{TK_SEM } \langle \text{stmts} \rangle \text{TK_END}$

This has no issue of violations in LL(1) compatibility. The nonterminal $\langle \text{input_par} \rangle$ is an essential construct to be part of the function definition while a function may or may not return values. Hence the $\langle \text{output_par} \rangle$ can have a syntax of (6) .

$$\text{FIRST}(\langle \text{function} \rangle) = \{\text{TK_FUNID}\}$$

5. $\langle \text{input_par} \rangle ::= \text{TK_INPUT } \text{TK_PARAMETER } \text{TK_LIST } \text{TK_SQL } \langle \text{parameter_list} \rangle \text{TK_SQR}$

Single production having no conflict has its FIRST set as given follows:

$$\text{FIRST}(\langle \text{input_par} \rangle) = \{\text{TK_INPUT}\}$$

6. $\langle \text{output_par} \rangle ::= \text{TK_OUTPUT } \text{TK_PARAMETER } \text{TK_LIST } \text{TK_SQL } \langle \text{parameter_list} \rangle \text{TK_SQR} \mid \text{eps}$

Presence of epsilon production makes us verify whether $\text{FIRST}(\alpha) \cap \text{FOLLOW}(\langle \text{output_par} \rangle) = \phi$ or not, where α represents the right hand side

$$\text{TK_OUTPUT } \text{TK_PARAMETER } \text{TK_LIST } \text{TK_SQL } \langle \text{parameter_list} \rangle \text{TK_SQR}$$

$$\text{FIRST}(\alpha) = \{\text{TK_OUTPUT}\}$$

Refer rule 4 to compute the $\text{FOLLOW}(\langle \text{output_par} \rangle)$ as below

$$\text{FOLLOW}(\langle \text{output_par} \rangle) = \{\text{TK_SEM}\}$$

$$\text{This implies that } \text{FIRST}(\alpha) \cap \text{FOLLOW}(\langle \text{output_par} \rangle) = \phi$$

Hence the given rules for $\langle \text{output_par} \rangle$ conform to the LL(1) specifications.

7. $\langle \text{parameter_list} \rangle ::= \langle \text{dataType} \rangle \text{TK_ID } \langle \text{remaining_list} \rangle$

The single production for $\langle \text{parameter_list} \rangle$ is trivially LL(1) compatible.

$$\text{FIRST}(\langle \text{parameter_list} \rangle) = \text{FIRST}(\langle \text{dataType} \rangle)$$

$$= \text{FIRST}(\langle \text{primitiveDatatype} \rangle) \cup \text{FIRST}(\langle \text{constructedDatatype} \rangle)$$

$$= \{\text{TK_INT}, \text{TK_REAL}, \text{TK_RECORD}, \text{TK_UNION}\}$$

Here we will allow the usage of record as well as union as parameters within the grammar as we did in this rule. This means that a function with following parameters will be considered as syntactically correct.

```
_recordDemo1 input parameter list [record #book d5cc34, union #newbook d2cd, int d3, real d5]
output parameter list[record #book d3];
```

But, a union variable passed as parameter or returned as a parameter will not make any sense due to the obvious reasons. However, this semantics will be captured in semantic analysis phase and you will have more clarity on type checking and semantic rules when your stage 2 work starts.

8. $\langle \text{dataType} \rangle \implies \langle \text{primitiveDatatype} \rangle \mid \langle \text{constructedDatatype} \rangle$

$$\begin{aligned} \text{FIRST}(\langle \text{primitiveDatatype} \rangle) \cap \text{FIRST}(\langle \text{constructedDatatype} \rangle) \\ = \{ \text{TK_INT}, \text{TK_REAL} \} \cap \{ \text{TK_RECORD}, \text{TK_UNION} \} \\ = \phi \end{aligned}$$

Since there is no nullable production for $\langle \text{dataType} \rangle$, there is no need for computing FOLLOW($\langle \text{dataType} \rangle$).

9. $\langle \text{primitiveDatatype} \rangle \implies \text{TK_INT} \mid \text{TK_REAL}$

$$\text{FIRST}(\text{TK_INT}) \cap \text{FIRST}(\text{TK_REAL}) = \{ \text{TK_INT} \} \cap \{ \text{TK_REAL} \} = \phi$$

Hence the rules for $\langle \text{primitiveDatatype} \rangle$ are LL(1) compatible.

$$\text{FIRST}(\langle \text{primitiveDatatype} \rangle) = \{ \text{TK_INT}, \text{TK_REAL} \}$$

10. $\langle \text{constructedDatatype} \rangle \implies \text{TK_RECORD TK_RUID} \mid \text{TK_UNION TK_RUID}$

This non-terminal supports the parameter list or the declaration statements, which use the record and union datatype. If these are redefined using the $\langle \text{defintype} \rangle$ or alias method as given below

```
defintype record #abc as #pqr
record #abc
    type int : x;
```

```

        type real:y;
    endrecord
    definetype record #definitionone as #definitiontwo
    record #definitionone
        type int : z
        type #pqr :a;          %non-recursive nested using definetype
        type record #abc: b;    %non-recursive nested without using definetype
        type #definitiontwo : c; %recursive nested using definetype
    endrecord

```

Consider the following declaration statements,

```

type record #definitionone c2d4;
type #definitiontwo c5; %using the alias of redefinition as above.

```

Let us look into the requirement of modification in rules for <constructedDatatype> from the perspective of two usages of data type-one in the declaration statement and the other in the parameter list. Both can use record and union data types to declare type of their variables/ parameters. Also both can use the alias or the type definition. The newly defined type is represented by the token TK_RUID.

Therefore the rules for <datatype> must include RUID in its first set also. The example in parameter list using both type names is as follows

```

_recordDemo1 input parameter list [ record #definitionone c2d4, #definitiontwo c5, record #abc d5cc34, #pqr d2cd]
output parameter list[record #abc d3];

```

Therefore, modify the rules for <constructedDatatype> as follows

$\langle \text{constructedDatatype} \rangle ::= \text{TK_RECORD TK_RUID} \mid \text{TK_UNION TK_RUID} \mid \text{TK_RUID}$ 10-a

$\text{FIRST}(\langle \text{constructedDatatype} \rangle) = \{\text{TK_RECORD}, \text{TK_UNION}, \text{TK_RUID}\}$

Also since the first sets of RHS of the three rules for $\langle \text{constructedDatatype} \rangle$ are disjoint, the grammar is LL(1).

That is $\text{FIRST}(\text{TK_RECORD TK_RUID}) \cap \text{FIRST}(\text{TK_UNION TK_RUID}) \cap \text{FIRST}(\text{TK_RUID})$

$= \{\text{TK_RECORD}\} \cap \{\text{TK_UNION}\} \cap \{\text{TK_RUID}\}$

$= \phi$

11. $\langle \text{remaining_list} \rangle ::= \text{TK_COMMA} \langle \text{parameter_list} \rangle \mid \text{eps}$

There exist a nullable production for the non-terminal $\langle \text{remaining_list} \rangle$. The RHS (right hand side) of the other production is such that the null string eps (epsilon) is not derivable from it.

Also

$\text{FIRST}(\text{TK_COMMA} \langle \text{parameter_list} \rangle) \cap \text{FOLLOW}(\langle \text{remaining_list} \rangle)$ should be empty.

$\text{FIRST}(\text{TK_COMMA} \langle \text{parameter_list} \rangle) = \{\text{TK_COMMA}\}$

and

$\text{FOLLOW}(\langle \text{remaining_list} \rangle) = \text{FOLLOW}(\text{parameter_list})$ from (7)

$= \{\text{TK_SQR}\}$

Hence,

$\text{FIRST}(\text{TK_COMMA} \langle \text{parameter_list} \rangle) \cap \text{FOLLOW}(\langle \text{remaining_list} \rangle)$

$= \{\text{TK_COMMA}\} \cap \{\text{TK_SQR}\}$

$= \phi$

Therefore both the given rules for the non-terminal $\langle \text{remaining_list} \rangle$ conform to LL(1) specifications.

12. $\langle \text{stmts} \rangle ::= \langle \text{typeDefinitions} \rangle \langle \text{declarations} \rangle \langle \text{otherStmts} \rangle \langle \text{returnStmt} \rangle$

The nonterminal $\langle \text{stmts} \rangle$ specifies the grammar for the body of the function. The ordering of other nonterminals such as $\langle \text{typeDefinitions} \rangle$, $\langle \text{declarations} \rangle$ and $\langle \text{returnStmt} \rangle$ have fixed positions within the body of the function code.

There is no epsilon production in $\langle \text{stmts} \rangle$ and a single non nullable production for $\langle \text{stmts} \rangle$ is LL(1) compatible.

In order to find the first set of $\langle \text{stmts} \rangle$, we need to union the first sets as follows. The non-terminals $\langle \text{typeDefinitions} \rangle$, $\langle \text{declarations} \rangle$ and $\langle \text{otherStmts} \rangle$ are nullable and have reachability through the application of their null productions.

$$\text{FIRST}(\langle \text{stmts} \rangle) = \text{FIRST}(\langle \text{typeDefinitions} \rangle) \cup \text{FIRST}(\langle \text{declarations} \rangle) \cup \text{FIRST}(\langle \text{otherStmts} \rangle) \cup \text{FIRST}(\langle \text{returnStmt} \rangle)$$

.....using 13, 19, 22 and 43

$$\begin{aligned} &= \{\text{TK_RECORD}, \text{TK_UNION}\} \cup \{\text{TK_TYPE}\} \cup \{\text{TK_ID}, \text{TK_RUID}, \text{TK_WHILE}, \text{TK_IF}, \text{TK_READ}, \text{TK_WRITE}, \text{TK_SQL}, \\ &\text{TK_CALL}\} \cup \{\text{TK_RETURN}\} \\ &= \{\text{TK_RECORD}, \text{TK_UNION}, \text{TK_TYPE}, \text{TK_ID}, \text{TK_RUID}, \text{TK_WHILE}, \text{TK_IF}, \text{TK_READ}, \text{TK_WRITE}, \text{TK_SQL}, \\ &\text{TK_CALL}, \text{TK_RETURN}\} \end{aligned}$$

Since the nonterminal $\langle \text{stmts} \rangle$ does not have a nullable production, we need not compute FOLLOW($\langle \text{stmts} \rangle$) for populating the parsing table. Recall the construction of the parsing table. The columns corresponding to the tokens in FIRST($\langle \text{stmts} \rangle$) corresponding to the row $\langle \text{stmts} \rangle$ are populated with the rule number 12.

13. $\langle \text{typeDefinitions} \rangle ::= \langle \text{typeDefinition} \rangle \langle \text{typeDefinitions} \rangle \mid \epsilon$

This is the place where the $\langle \text{definetypestmt} \rangle$ finds its existence. As it is used for record and union types redefinitions (or alias). Let us have this construct appear anywhere within the piece of code reserved for $\langle \text{typeDefinitions} \rangle$

Redefine the rule 13

$$\langle \text{typeDefinitions} \rangle ::= \langle \text{actualOrRedefined} \rangle \langle \text{typeDefinitions} \rangle \mid \epsilon \quad \dots 13\text{-a}$$

Where

$$\langle \text{actualOrRedefined} \rangle ::= \langle \text{typeDefinition} \rangle \mid \langle \text{definetypestmt} \rangle \quad \dots 13\text{-b}$$

With this, the constructs $\langle \text{typeDefinition} \rangle$ and $\langle \text{definetypestmt} \rangle$ can appear in any order before the $\langle \text{stmts} \rangle$ construct within a function.

The semantic analyzer will be able to catch the relevance of the presence of a type redefinition and type equivalence later. If your test case has the following code

```

definetype union #old as #new % line 1- derived using <definetypestmt>.
% following record type definition is derived using <typeDefinition>
record #definitionone
    type int : z
    type #pqr :a; %non-recursive nested using definetype
    type record #abc: b; %non-recursive nested without using definetype
    type #definitiontwo : c; % line 2
endrecord
definetype record #definitionone as #definitiontwo % derived using <definetypestmt>.

```

Note that here the record #old in line 1 is not defined using <typeDefinition> grammar but will still be syntactically correct. The semantic analyzer will be able to catch this error. Also note that the record #definitionone is redefined as #definitiontwo even after its use above in line - 2. The semantic analyzer will require two pass mechanism to collect all information.

Let us analyze 13-b first.

Since $\text{FIRST}(\langle \text{typeDefinition} \rangle) \cap \text{FIRST}(\langle \text{definetypestmt} \rangle)$

$= \{\text{TK_RECORD}, \text{TK_UNION}\} \cap \{\text{TK_DEFINETYPE}\}$ using 14, 15 and 46

Therefore rule 13-b is LL(1) compatible.

And $\text{FIRST}(\langle \text{actualOrRedefined} \rangle) = \{\text{TK_RECORD}, \text{TK_UNION}, \text{TK_DEFINETYPE}\}$

Now 13-a

$\text{FIRST}(\langle \text{actualOrRedefined} \rangle) \cap \text{FOLLOW}(\langle \text{typeDefinitions} \rangle)$

$= \{\text{TK_RECORD}, \text{TK_UNION}, \text{TK_DEFINETYPE}\} \cap \text{FIRST}(\langle \text{declarations} \rangle)$

$$\begin{aligned}
&= \{TK_RECORD, TK_UNION, TK_DEFINETYPE\} \cap FIRST(<declaration>) \\
&= \{TK_RECORD, TK_UNION, TK_DEFINETYPE\} \cap \{TK_TYPE\} \\
&= \phi
\end{aligned}$$

Therefore, rule 13-a is LL(1) compatible

14. $\langle typeDefinition \rangle ::= TK_RECORD \ TK_RUID \ \langle fieldDefinitions \rangle \ TK_ENDRECORD$

15. $\langle typeDefinition \rangle ::= TK_UNION \ TK_RUID \ \langle fieldDefinitions \rangle \ TK_ENDUNION$

The rule 14 reaches to the

$FIRST(TK_RECORD \ TK_RUID \ \langle fieldDefinitions \rangle \ TK_ENDRECORD) = \{TK_RECORD\}$

Similarly, rule 15 reaches the first of its RHS

$FIRST(TK_UNION \ TK_RUID \ \langle fieldDefinitions \rangle \ TK_ENDUNION) = \{TK_UNION\}$

Also, the Firsts sets of the RHS of the two rules 14 and 15 for the non-terminal $\langle typeDefinition \rangle$ are disjoint.

$\{TK_RECORD\} \cup \{TK_UNION\} = \phi$

Therefore rules 14 and 15 conform to the LL(1) specifications and do not require any modification.

Also, $FIRST(\langle typeDefinition \rangle) = \{TK_RECORD, TK_UNION\}$

16. $\langle fieldDefinitions \rangle ::= \langle fieldDefinition \rangle \langle fieldDefinition \rangle \langle moreFields \rangle$

As two consecutive occurrences of $\langle fieldDefinition \rangle$ are there at the RHS of the above production, this imposes a requirement of at least two fields in a record. There can be one or more fields afterwards. As the above rule is single and non nullable, it is LL(1) compatible.

$FIRST(\langle fieldDefinitions \rangle) = FIRST(\langle fieldDefinition \rangle) = \{TK_TYPE\}$

17. $\langle fieldDefinition \rangle ::= TK_TYPE \ \langle primitiveDatatype \rangle \ TK_COLON \ TK_FIELDID \ TK_SEM$

Here, the type of the field needs major change to support the nested record and union data types. As per the descriptions given in the language specification document, a field name can also be of type union or record. However, only the alias of the union or record type is allowed in the type

definition of these fields. A minor correction in the example given in the language specifications document that uses type newname: field should be updated using a # as prefix. The updated example is given below.

```
defintype union #student as #newname
record #taggedunionexample
    type int : tagvalue;
    type #newname : field; %add # before newname which was missing in the document earlier.
endrecord
```

Therefore the rule for <fielddefinition> is modified as given below.

<fieldDefinition>====> TK_TYPE <fieldType> TK_COLON TK_FIELDID TK_SEM

<fieldtype>====> <primitiveDatatype> | TK_RUID

..... Add this rule in the grammar 17(a)

Note that the record or union data type can be made nested by using the alias. This grammar will not support the usage of usual type definition syntax of union and record to be nested here. To explain this, the following example **will not be syntactically valid**.

```
record #taggedunionexample
    type int : tagvalue;
    type union #student
        type int:rollno;
        type real marks;
        type int age;
    endunion : field; %add # before newname which was missing in the document earlier.
endunion
endrecord
```

%NOT CORRECT SYNTACTICALLY

Now check the compatibility of the two rules in 17(a)

$\text{FIRST}(\text{<primitiveDatatype>}) \cap \text{FIRST}(\text{TK_RUID}) = \{\text{TK_INT}, \text{TK_REAL}\} \cap \{\text{TK_RUID}\} = \phi$

And $\text{FIRST}(\text{<fieldType>}) = \{\text{TK_INT}, \text{TK_REAL}, \text{TK_RUID}\}$

$\text{FIRST}(\langle \text{fieldDefinition} \rangle) = \{\text{TK_TYPE}\}$

18. $\langle \text{moreFields} \rangle \Rightarrow \langle \text{fieldDefinition} \rangle \langle \text{moreFields} \rangle \mid \epsilon$

Let us find

$\text{FOLLOW}(\langle \text{moreFields} \rangle) = \phi \cup \text{FOLLOW}(\langle \text{fieldDefinitions} \rangle) = \{\text{TK_ENDRECORD}, \text{TK_ENDUNION}\}$ using rules 14, 15 and 16

$$\begin{aligned} & \text{FIRST}(\langle \text{fieldDefinition} \rangle \langle \text{moreFields} \rangle) \cap \text{FOLLOW}(\langle \text{moreFields} \rangle) \\ &= \{\text{TK_TYPE}\} \cap \{\text{TK_ENDRECORD}, \text{TK_ENDUNION}\} \\ &= \phi \end{aligned}$$

Hence, the above two rules for the non terminal $\langle \text{moreFields} \rangle$ are LL(1) compatible.

19. $\langle \text{declarations} \rangle \Rightarrow \langle \text{declaration} \rangle \langle \text{declarations} \rangle \mid \epsilon$

$$\begin{aligned} \text{FOLLOW}(\langle \text{declarations} \rangle) &= \text{FIRST}(\langle \text{otherStmts} \rangle) \cup \text{FOLLOW}(\langle \text{otherStmts} \rangle) \\ &= \{\text{TK_ID}, \text{TK_RUID}, \text{TK_WHILE}, \text{TK_IF}, \text{TK_READ}, \text{TK_WRITE}, \text{TK_SQL}, \text{TK_CALL}\} \end{aligned}$$

$$\text{FIRST}(\langle \text{declaration} \rangle \langle \text{declarations} \rangle) = \text{FIRST}(\langle \text{declaration} \rangle) = \{\text{TK_TYPE}\}$$

We observe that both of the above sets are disjoint. Therefore both the above productions for the non terminal $\langle \text{declarations} \rangle$ conform to LL(1) specifications.

20. $\langle \text{declaration} \rangle \Rightarrow \text{TK_TYPE} \langle \text{dataType} \rangle \text{TK_COLON} \text{TK_ID} \text{TK_COLON} \langle \text{global_or_not} \rangle \text{TK_SEM}$

$\langle \text{declaration} \rangle \Rightarrow \text{TK_TYPE} \langle \text{dataType} \rangle \text{TK_COLON} \text{TK_ID} \langle \text{global_or_not} \rangle \text{TK_SEM}$

Also to incorporate the optional existence of global keyword prefixed with a colon is reflected as modification in rule 21.

$\text{FIRST}(\langle \text{declaration} \rangle) = \{\text{TK_TYPE}\}$

Rule conforms to LL(1) specifications.

21. $\langle \text{global_or_not} \rangle \Rightarrow \text{TK_GLOBAL} \mid \epsilon$

This rule is also modified as follows

$\langle \text{global_or_not} \rangle \Rightarrow \text{TK_COLON TK_GLOBAL} \mid \text{eps}$

$\text{FOLLOW}(\langle \text{global_or_not} \rangle) = \{\text{TK_SEM}\}$

.....using 20

and

$\text{FIRST}(\text{TK_COLON TK_GLOBAL}) = \{\text{TK_COLON}\}$

As

$\text{FOLLOW}(\langle \text{global_or_not} \rangle) \cap \text{FIRST}(\text{TK_COLON TK_GLOBAL}) = \{\text{TK_SEM}\} \cap \{\text{TK_COLON}\} = \emptyset$

Therefore the new rule for $\langle \text{global_or_not} \rangle$ is LL(1) compatible.

22. $\langle \text{otherStmts} \rangle \Rightarrow \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \mid \text{eps}$

$\text{FOLLOW}(\langle \text{otherStmts} \rangle) = \text{FIRST}(\langle \text{returnStmts} \rangle)$

$\text{FIRST}(\langle \text{otherStmts} \rangle) = \text{FIRST}(\langle \text{stmt} \rangle)$

[Note: $\langle \text{stmt} \rangle$ does not derive an empty string]

$= \{\text{TK_ID}, \text{TK_WHILE}, \text{TK_IF}, \text{TK_READ}, \text{TK_WRITE}, \text{TK_SQL}, \text{TK_CALL}\}$

.....using 23

23. $\langle \text{stmt} \rangle \Rightarrow \langle \text{assignmentStmt} \rangle \mid \langle \text{iterativeStmt} \rangle \mid \langle \text{conditionalStmt} \rangle \mid \langle \text{ioStmt} \rangle \mid \langle \text{funCallStmt} \rangle$

Referring rules 24, 29, 30, 32 and 26, we conclude that the five productions given above for the nonterminal $\langle \text{stmt} \rangle$ are not nullable.

We must check the LL(1) property that the FIRST sets of the RHSs of all productions above are disjoint.

$\text{FIRST}(\langle \text{assignmentStmt} \rangle) \cap \text{FIRST}(\langle \text{iterativeStmt} \rangle) \cap \text{FIRST}(\langle \text{conditionalStmt} \rangle) \cap \text{FIRST}(\langle \text{ioStmt} \rangle) \cap \text{FIRST}(\langle \text{funCallStmt} \rangle)$

$= \{\text{TK_ID}\} \cap \{\text{TK_WHILE}\} \cap \{\text{TK_IF}\} \cap \{\text{TK_READ}, \text{TK_WRITE}\} \cap \{\text{TK_SQL}, \text{TK_CALL}\}$

$= \emptyset$

Hence, the above five productions for the nonterminal $\langle \text{stmt} \rangle$ conform to LL(1) specifications.

and ,

$\text{FIRST}(\langle \text{stmt} \rangle) = \{\text{TK_ID}, \text{TK_WHILE}, \text{TK_IF}, \text{TK_READ}, \text{TK_WRITE}, \text{TK_SQL}, \text{TK_CALL}\}$

24. $\langle \text{assignmentStmt} \rangle \Rightarrow \langle \text{SingleOrRecId} \rangle \text{TK_ASSIGNOP} \langle \text{arithmeticExpression} \rangle \text{TK_SEM}$

$\text{FIRST}(\langle \text{assignmentStmt} \rangle) = \text{FIRST}(\langle \text{singleOrRecId} \rangle) = \{\text{TK_ID}\}$

.....using 25 e

25. $\langle \text{singleOrRecId} \rangle ::= \text{TK_ID} \mid \text{TK_ID TK_DOT TK_FIELDID}$

The second rule for $\langle \text{singleOrRecId} \rangle$ does not support the record or union type variable construction expanded using the dot operator for any number of times. As you have seen in the examples given under rules for $\langle \text{constructedDatatype} \rangle$, if a variable is declared as follows

```
definetype record #definitionone as #definitiontwo
  record #definitionone
    type int : z
    type #pqr : a;           %non-recursive nested using definetype
    type record #abc: b;     %non-recursive nested without using definetype
    type #definitiontwo : c; %recursive nested using definetype
  endrecord
```

Consider the following declaration statements,

```
type record #definitionone c2d4;
type #definitiontwo c5;
```

and different variables constructed as c2d4.c, c5.b, c2cd4.c.c.z, c2cd4.c.c.c.c.c.z, and so on.

Then the following rules for $\langle \text{singleOrRecId} \rangle$ will not help as the token stream would then be TK_ID TK_DOT TK_FIELDID TK_DOT TK_FIELDID TK_DOT TK_FIELDID TK_DOT TK_FIELDID TK_DOT TK_FIELDID TK_DOT TK_FIELDID TK_DOT TK_FIELDID for a variable c2cd4.c.c.c.c.c.z (for example)

$\langle \text{singleOrRecId} \rangle ::= \text{TK_ID} \mid \text{TK_ID TK_DOT TK_FIELDID}$

So we need to redesign the rule

$\langle \text{singleOrRecId} \rangle ::= \text{TK_ID TK_DOT TK_FIELDID}$

As

$\langle \text{singleOrRecId} \rangle \implies \langle \text{constructedVariable} \rangle$

by introducing a new nonterminal say $\langle \text{constructedVariable} \rangle$ which will represent the variables as described above.

Therefore the rules for $\langle \text{singleOrRecId} \rangle$ will become

$\langle \text{singleOrRecId} \rangle \implies \text{TK_ID} \mid \langle \text{constructedVariable} \rangle$ 25 a

We can define a recursive rule for this as follows.

$\langle \text{constructedVariable} \rangle \implies \text{TK_ID} \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$ -----25 b

$\langle \text{oneExpansion} \rangle \implies \text{TK_DOT TK_FIELDID}$ -----25 c

$\langle \text{moreExpansions} \rangle \implies \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle \mid \epsilon$ -----25 d

Here two more non-terminals are introduced to address nested record types for the variables constructed using them. It is clear that the nesting level of zero, i.e. with no record type field, the nonterminal $\langle \text{oneExpansion} \rangle$ derives TK_DOT TK_FIELDID and in this case the null production for $\langle \text{moreExpansions} \rangle$ is used. For nested definitions, a number of times $\langle \text{moreExpansions} \rangle \implies \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$ rule is applied as required..

If you wish you can replace the name $\langle \text{singleOrRecId} \rangle$ with $\langle \text{singleOrRecOrUnionID} \rangle$ everywhere to make it self-explanatory.

Before going for verifying the LL(1) compatibility of the rule for $\langle \text{singleOrRecId} \rangle$, let us examine the newly introduced rules.

Rule 25 b is a single rule and is non nullable.

$\text{FIRST}(\langle \text{constructedVariable} \rangle) = \{\text{TK_ID}\}$

$\text{FOLLOW}(\langle \text{constructedVariable} \rangle) = \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) = \{\text{TK_ASSIGNOP}\}$

Rule 25 c is LL(1) trivially.

$$\text{FIRST}(\langle \text{oneExpansion} \rangle) = \{\text{TK_DOT}\}$$

Rule 25 d requires that $\text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle)$ and $\text{FOLLOW}(\langle \text{moreExpansions} \rangle)$ should be disjoint.

$$\begin{aligned} & \text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle) \cap \text{FOLLOW}(\langle \text{moreExpansions} \rangle) \\ = & \{\text{TK_DOT}\} \cap \text{FOLLOW}(\langle \text{constructedVariable} \rangle) \\ = & \{\text{TK_DOT}\} \cap \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) \\ = & \{\text{TK_DOT}\} \cap \{\text{TK_ASSIGNOP}, \text{TK_CL}\} \\ = & \phi \end{aligned}$$

Using 25 a and 25 b

$$\langle \text{singleOrRecId} \rangle ::= \text{TK_ID} \mid \text{TK_ID} \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle$$

Both of the above productions for the nonterminal $\langle \text{singleOrRecId} \rangle$ are non nullable and their first sets are not disjoint. This requires left factoring

The above two rules are left factored using new non terminal $\langle \text{new_24} \rangle$

$$\langle \text{singleOrRecId} \rangle ::= \text{TK_ID} \langle \text{option_single_constructed} \rangle \quad \text{.....25 e}$$

$$\langle \text{option_single_constructed} \rangle ::= \epsilon \mid \langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle \quad \text{.....25 f}$$

25 e is trivially LL(1) compatible.

For both rules for $\langle \text{option_single_constructed} \rangle$ in 25 f, $\text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle)$ and $\text{FOLLOW}(\langle \text{option_single_constructed} \rangle)$ should be disjoint.

$$\begin{aligned} & \text{FIRST}(\langle \text{oneExpansion} \rangle \langle \text{moreExpansions} \rangle) \cap \text{FOLLOW}(\langle \text{option_single_constructed} \rangle) \\ = & \{\text{TK_DOT}\} \cap \text{FOLLOW}(\langle \text{singleOrRecId} \rangle) \\ = & \{\text{TK_DOT}\} \cap \{\text{TK_ASSIGNOP}, \text{TK_CL}\} \end{aligned}$$

$= \phi$.

Therefore, the above productions are LL(1) compatible.

The final rules for $\langle \text{singleOrRecId} \rangle$ include 25 e, f, c and d.

Using 25 e, $\text{FIRST}(\langle \text{singleOrRecId} \rangle) = \{\text{TK_ID}\}$

26. $\langle \text{funCallStmt} \rangle ::= \langle \text{outputParameters} \rangle \text{ TK_CALL TK_FUNID TK_WITH TK_PARAMETERS } \langle \text{inputParameters} \rangle$

The rule must have at its end TK_SEM to represent semicolon

$\langle \text{funCallStmt} \rangle ::= \langle \text{outputParameters} \rangle \text{ TK_CALL TK_FUNID TK_WITH TK_PARAMETERS } \langle \text{inputParameters} \rangle \text{ TK_SEM}$

The rule is trivially LL(1) compatible.

$\text{FIRST}(\langle \text{funCallStmt} \rangle) = \text{FIRST}(\langle \text{outputParameters} \rangle) \cup \text{FOLLOW}(\langle \text{outputParameters} \rangle)$
 $= \{\text{TK_SQL}\} \cup \{\text{TK_CALL}\} = \{\text{TK_SQL}, \text{TK_CALL}\}$

27. $\langle \text{outputParameters} \rangle ::= \text{TK_SQL } \langle \text{idList} \rangle \text{ TK_SQR TK_ASSIGNOP } | \in$

$\text{FOLLOW}(\langle \text{outputParameters} \rangle) = \{\text{TK_CALL}\}$

Due to the presence of the nullable production for the non terminal $\langle \text{outputParameters} \rangle$, we check the following LL(1) property

$\text{FIRST}(\text{TK_SQL } \langle \text{idList} \rangle \text{ TK_SQR TK_ASSIGNOP}) \cap \text{FOLLOW}(\langle \text{outputParameters} \rangle)$
 $= \{\text{TK_SQL}\} \cap \{\text{TK_CALL}\} = \phi$

Hence, the productions in 27 are LL(1) compatible.

$\text{FIRST}(\langle \text{outputParameters} \rangle) = \{\text{TK_SQL}\}$

28. $\langle \text{inputParameters} \rangle ::= \text{TK_SQL } \langle \text{idList} \rangle \text{ TK_SQR}$

LL(1) compatible with $\text{FIRST}(\langle \text{inputParameters} \rangle) = \{\text{TK_SQL}\}$

29. $\langle \text{iterativeStmt} \rangle ::= \text{TK_WHILE TK_OP } \langle \text{booleanExpression} \rangle \text{ TK_CL } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK_ENDWHILE}$

LL(1) compatible with $\text{FIRST}(\langle \text{iterativeStmt} \rangle) = \{\text{TK_WHILE}\}$

30. $\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK_IF } \langle \text{booleanExpression} \rangle \text{ TK_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK_ELSE } \langle \text{otherStmts} \rangle \text{ TK_ENDIF}$

31. $\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK_IF } \langle \text{booleanExpression} \rangle \text{ TK_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK_ENDIF}$

The two rules (30) and (31) for the nonterminal $\langle \text{conditionalStmt} \rangle$ need modifications as they originally form ambiguous grammar. We need to perform the left factoring as follows.

$\langle \text{conditionalStmt} \rangle \Rightarrow \text{TK_IF } \text{TK_OP } \langle \text{booleanExpression} \rangle \text{ TK_CL TK_THEN } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \langle \text{elsePart} \rangle$ number rule as 30-a
 $\langle \text{elsePart} \rangle \Rightarrow \text{TK_ELSE } \langle \text{stmt} \rangle \langle \text{otherStmts} \rangle \text{ TK_ENDIF} \mid \text{TK_ENDIF}$ number rules as 31-a

Notice that I also added TK_OP and TK_CL in new rule 30-a to enclose boolean expression. These two tokens were missing earlier.

Also I am putting a constraint of at least one statement within else part.

Now there is a single rule for $\langle \text{conditionalStmt} \rangle$ which is non nullable and its first set is $\{\text{TK_IF}\}$

But the newly introduced nonterminal $\langle \text{elsePart} \rangle$ has two non nullable productions and their corresponding FIRST sets are disjoint.

$\text{FIRST}(\text{TK_ELSE } \langle \text{otherStmts} \rangle \text{ TK_ENDIF}) \cap \text{FIRST}(\text{TK_ENDIF}) = \{\text{TK_ELSE}\} \cap \{\text{TK_ENDIF}\} = \phi$

Therefore the new set of rules for $\langle \text{elsePart} \rangle$ is also LL(1) compatible.

And,

$\text{FIRST}(\langle \text{conditionalStmt} \rangle) = \{\text{TK_IF}\}$

$\text{FIRST}(\langle \text{elsePart} \rangle) = \{\text{TK_ELSE}, \text{TK_ENDIF}\}$

32. $\langle \text{ioStmt} \rangle \Rightarrow \text{TK_READ TK_OP } \langle \text{var} \rangle \text{ TK_CL TK_SEM} \mid \text{TK_WRITE TK_OP } \langle \text{var} \rangle \text{ TK_CL TK_SEM}$

Both the above productions are non nullable.

and,

$\text{FIRST}(\text{TK_READ TK_OP } \langle \text{var} \rangle \text{ TK_CL TK_SEM}) \cap \text{FIRST}(\text{TK_WRITE TK_OP } \langle \text{var} \rangle \text{ TK_CL TK_SEM})$

$$= \{TK_READ\} \cap \{TK_WRITE\} = \phi$$

Therefore, the above productions are LL(1) compatible.

and,

$$FIRST(<ioStmt>) = \{TK_READ, TK_WRITE\}$$

33. $\langle arithmeticExpression \rangle \implies \langle arithmeticExpression \rangle \langle operator \rangle \langle arithmeticExpression \rangle$

I had left a major correction for you to do in the arithmetic expression grammar. You were required to impose precedence of operators and the operations on record variables.

34. $\langle arithmeticExpression \rangle \implies TK_OP \langle arithmeticExpression \rangle TK_CL \mid \langle var \rangle$

An arithmetic expression is expected to take as argument an identifier, statically available integer number, real number, a record identifier and a record variable identifier (constructed by expanding the name using dot with a field name). These are represented in the non-terminal $\langle var \rangle$ in rule number 39-a.

Now we reformulate the rules 33, 34 and 35 to impose precedence.

$\langle arithmeticExpression \rangle \implies \langle arithmeticExpression \rangle \langle lowPrecedenceOperators \rangle \langle term \rangle \mid \langle term \rangle$ (A1)

$\langle term \rangle \implies \langle term \rangle \langle highPrecedenceOperators \rangle \langle factor \rangle \mid \langle factor \rangle$ (A2)

$\langle factor \rangle \implies TK_OP \langle arithmeticExpression \rangle TK_CL \mid \langle var \rangle$ (A3)

$\langle highPrecedenceOperator \rangle \implies TK_MUL \mid TK_DIV$ (A4)

$\langle lowPrecedenceOperators \rangle \implies TK_PLUS \mid TK_MINUS$ (A5)

[Note : I have given new numbers for referring to the rules, but you can have your own numbering scheme for the rules]

Now let us verify the rules A1-5 for the nonterminals $\langle arithmeticExpression \rangle$, $\langle term \rangle$, $\langle factor \rangle$, $\langle lowPrecedenceOperators \rangle$, $\langle highPrecedenceOperators \rangle$,

The rule A1 needs left recursion elimination and the new set of productions are obtained by introducing new nonterminal, say $\langle expPrime \rangle$

$\langle arithmeticExpression \rangle \implies \langle term \rangle \langle expPrime \rangle$

$\langle expPrime \rangle \implies \langle lowPrecedenceOperators \rangle \langle term \rangle \langle expPrime \rangle \mid \epsilon$

Similarly rule A2 also needs left recursion elimination. The new set of productions are obtained by introducing new nonterminal, say, $\langle \text{termPrime} \rangle$

$$\langle \text{term} \rangle \Rightarrow \langle \text{factor} \rangle \langle \text{termPrime} \rangle$$

$$\langle \text{termPrime} \rangle \Rightarrow \langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle \mid \epsilon$$

Now we have two more nonterminals $\langle \text{expPrime} \rangle$ and $\langle \text{termPrime} \rangle$. The complete set of non left recursive, unambiguous arithmetic expression grammar then becomes with new numbers B1-B7

- B1) $\langle \text{arithmeticExpression} \rangle \Rightarrow \langle \text{term} \rangle \langle \text{expPrime} \rangle$
 B2) $\langle \text{expPrime} \rangle \Rightarrow \langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle \mid \epsilon$
 B3) $\langle \text{term} \rangle \Rightarrow \langle \text{factor} \rangle \langle \text{termPrime} \rangle$
 B4) $\langle \text{termPrime} \rangle \Rightarrow \langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle \mid \epsilon$
 B5) $\langle \text{factor} \rangle \Rightarrow \text{TK_OP} \langle \text{arithmeticExpression} \rangle \text{TK_CL} \mid \langle \text{var} \rangle$
 B6) $\langle \text{highPrecedenceOperator} \rangle \Rightarrow \text{TK_MUL} \mid \text{TK_DIV}$
 B7) $\langle \text{lowPrecedenceOperators} \rangle \Rightarrow \text{TK_PLUS} \mid \text{TK_MINUS}$

Analysis of B1: As there is only single rule available for the nonterminal $\langle \text{arithmeticExpression} \rangle$, we must see that is not nullable

$\text{FIRST}(\langle \text{arithmeticExpression} \rangle) = \text{FIRST}(\langle \text{term} \rangle) = \text{FIRST}(\langle \text{factor} \rangle) = \{\text{TK_ID}, \text{TK_NUM}, \text{TK_RNUM}, \text{TK_OP}\}$

There is no need to go further to check LL(1) compatibility. The grammar for $\langle \text{arithmeticExpression} \rangle$ is LL(1) compatible.

Analysis of B2: One of the productions for the nonterminal $\langle \text{expPrime} \rangle$ is nullable.

Let us verify $\text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle)$ and $\text{FOLLOW}(\langle \text{expPrime} \rangle)$ are disjoint.

$\text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle) = \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle)$
 $= \{\text{TK_PLUS}, \text{TK_MINUS}\}$ using analysis of B7

and,

$\text{FOLLOW}(\langle \text{expPrime} \rangle) = \text{FOLLOW}(\langle \text{arithmeticExpression} \rangle)$ using B1
 $= \{\text{TK_SEM}\}$ using 24

This implies that

$$\begin{aligned} & \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle \langle \text{term} \rangle \langle \text{expPrime} \rangle) \cap \text{FOLLOW}(\langle \text{expPrime} \rangle) \\ &= \{\text{TK_PLUS}, \text{TK_MINUS}\} \cap \{\text{TK_SEM}\} \\ &= \emptyset \end{aligned}$$

Therefore, B2 is also LL(1) compatible.

Analysis of B3:

As there is only single rule available for the nonterminal $\langle \text{term} \rangle$, we must see that is not nullable

$$\begin{aligned} & \text{FIRST}(\langle \text{term} \rangle) \\ &= \text{FIRST}(\langle \text{factor} \rangle) \\ &= \{\text{TK_ID}, \text{TK_NUM}, \text{TK_RNUM}, \text{TK_OP}\} \quad \text{.....using analysis of B5} \end{aligned}$$

There is no need to go further to check LL(1) compatibility. The grammar for $\langle \text{term} \rangle$ is LL(1) compatible.

Analysis of B4:

One of the productions for the nonterminal $\langle \text{termPrime} \rangle$ is nullable.

Let us verify $\text{FIRST}(\langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle)$ and $\text{FOLLOW}(\langle \text{termPrime} \rangle)$ are disjoint.

$$\begin{aligned} \text{FIRST}(\langle \text{highPrecedenceOperators} \rangle \langle \text{factor} \rangle \langle \text{termPrime} \rangle) &= \text{FIRST}(\langle \text{highPrecedenceOperators} \rangle) \\ &= \{\text{TK_MUL}, \text{TK_DIV}\} \quad \text{.....using analysis of B6} \end{aligned}$$

And,

$$\begin{aligned} \text{FOLLOW}(\langle \text{termPrime} \rangle) &= \text{FOLLOW}(\langle \text{term} \rangle) && \text{.....using B3} \\ &= \text{FIRST}(\langle \text{expPrime} \rangle) \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) && \text{.....using B2} \\ &= \text{FIRST}(\langle \text{lowPrecedenceOperators} \rangle) \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) && \text{.....using B2} \\ &= \{\text{TK_PLUS}, \text{TK_MINUS}\} \cup \text{FOLLOW}(\langle \text{expPrime} \rangle) && \text{.....using analysis of B7} \end{aligned}$$

$$\begin{aligned}
&= \{TK_PLUS, TK_MINUS\} \cup FOLLOW(<arithmeticExpression>) && \text{....using B1} \\
&= \{TK_PLUS, TK_MINUS\} \cup \{TK_SEM\} && \text{....using 24} \\
&= \{TK_PLUS, TK_MINUS, TK_SEM\}
\end{aligned}$$

Therefore,

$$\begin{aligned}
&FIRST(<highPrecedenceOperators><factor> <termPrime>) \cap FOLLOW(<termPrime>) \\
&= \{TK_MUL, TK_DIV\} \cap \{TK_PLUS, TK_MINUS, TK_SEM\} \\
&= \phi
\end{aligned}$$

Hence the rules defined by B4 conform to LL(1) specifications.

Analysis of B5: There are two productions for the non terminal <factor>.

$$FIRST(TK_OP <arithmeticExpression> TK_CL) = \{TK_OP\}$$

and,

$$FIRST(<var>) = \{TK_ID, TK_NUM, TK_RNUM\}$$

It is now obvious that no production for <factor> is a nullable production and,

$$FIRST(TK_OP <arithmeticExpression> TK_CL) \cap FIRST(<var>) = \phi$$

Therefore the grammar for <factor> is LL(1) compatible.

and,

$$FIRST(<factor>) = \{TK_ID, TK_NUM, TK_RNUM, TK_OP\}$$

Analysis of B6: The two productions of the nonterminal <highPrecedenceOperators> are trivially LL(1) compatible

$$FIRST(<highPrecedenceOperators>) = \{TK_MUL, TK_DIV\}$$

Analysis of B7: The two productions of the nonterminal <lowPrecedenceOperators> are also trivially LL(1) compatible

$$FIRST(<lowPrecedenceOperators>) = \{TK_PLUS, TK_MINUS\}$$

35. <operator> ==> TK_PLUS | TK_MUL |TK_MINUS|TK_DIV

This rule is discarded now.

36. $\langle \text{booleanExpression} \rangle \Rightarrow \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL} \langle \text{logicalOp} \rangle \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL}$
 $\text{FIRST}(\text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL} \langle \text{logicalOp} \rangle \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL}) = \{ \text{TK_OP} \}$
 and the production is not nullable.

37. $\langle \text{booleanExpression} \rangle \Rightarrow \langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle$
 $\text{FIRST}(\langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle) = \text{FIRST}(\langle \text{var} \rangle)$
 $= \{ \text{TK_ID}, \text{TK_NUM}, \text{TK_RNUM} \}$
 and the production is not nullable.

38. $\langle \text{booleanExpression} \rangle \Rightarrow \text{TK_NOT} \langle \text{booleanExpression} \rangle$

Introduce the parentheses pair around the $\langle \text{booleanExpression} \rangle$ as follows

$\langle \text{booleanExpression} \rangle \Rightarrow \text{TK_NOT} \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL}$ 38-a

Considering the above rules 36-38 for analysis, we observe that none of the three productions for the nonterminal $\langle \text{booleanExpression} \rangle$ is nullable (using First set of $\langle \text{var} \rangle$ from 39)

and, The first sets of the right hand sides of the three productions are disjoint.

$\text{FIRST}(\text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL} \langle \text{logicalOp} \rangle \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL})$
 $\cap \text{FIRST}(\langle \text{var} \rangle \langle \text{relationalOp} \rangle \langle \text{var} \rangle)$
 $\cap \text{FIRST}(\text{TK_NOT} \text{TK_OP} \langle \text{booleanExpression} \rangle \text{TK_CL})$
 $= \{ \text{TK_OP} \} \cap \{ \text{TK_ID}, \text{TK_NUM}, \text{TK_RNUM} \} \cap \{ \text{TK_NOT} \}$
 $= \phi$

Hence, the rules 36, 37 and 38-a for the nonterminal $\langle \text{booleanExpression} \rangle$ are LL(1) compatible.

39. $\langle \text{var} \rangle \Rightarrow \text{TK_ID} \mid \text{TK_NUM} \mid \text{TK_RNUM}$

This rule needs modifications. The non-terminal $\langle \text{var} \rangle$ participates in read and write statements and in arithmetic and Boolean expressions. It is expected to represent the variable identifier, the record and union variables constructed by expanding the identifier name with dot operator followed by the field name. Also $\langle \text{var} \rangle$ represents the statically available integer and real numbers.

The rule for $\langle \text{var} \rangle$ is modified as follows

$\langle \text{var} \rangle \Rightarrow \langle \text{singleOrRecId} \rangle \mid \text{TK_NUM} \mid \text{TK_RNUM}$

.....39 a.

The three rules on the RHS are LL(1) compatible as

$\text{FIRST}(\langle \text{singleOrRecId} \rangle) \cap \text{FIRST}(\text{TK_NUM}) \cap \text{FIRST}(\text{TK_RNUM}) = \phi$

And

$\text{FIRST}(\langle \text{var} \rangle) = \{\text{TK_ID}, \text{TK_NUM}, \text{TK_RNUM}\}$

40. $\langle \text{logicalOp} \rangle \Rightarrow \text{TK_AND} \mid \text{TK_OR}$

I have not defined the precedence of AND over OR or vice versa. We have the support to enclose the boolean expression in parentheses.

Trivially the productions for $\langle \text{logicalOp} \rangle$ conform to LL(1) specifications

41. $\langle \text{relationalOp} \rangle \Rightarrow \text{TK_LT} \mid \text{TK_LE} \mid \text{TK_EQ} \mid \text{TK_GT} \mid \text{TK_GE} \mid \text{TK_NE}$

All six productions above for the nonterminal $\langle \text{relationalOp} \rangle$ are trivially LL(1) compatible and

$\text{FIRST}(\langle \text{relationalOp} \rangle) = \{\text{TK_LT}, \text{TK_LE}, \text{TK_EQ}, \text{TK_GT}, \text{TK_GE}, \text{TK_NE}\}$

42. $\langle \text{returnStmt} \rangle \Rightarrow \text{TK_RETURN} \langle \text{optionalReturn} \rangle \text{TK_SEM}$

$\text{FIRST}(\langle \text{returnStmt} \rangle) = \{\text{TK_RETURN}\}$

and the production is not nullable, hence the rule is LL(1) compatible.

43. $\langle \text{optionalReturn} \rangle \Rightarrow \text{TK_SQL} \langle \text{idList} \rangle \text{TK_SQR} \mid \text{eps}$

$\text{FIRST}(\text{TK_SQL} \langle \text{idList} \rangle \text{TK_SQR}) = \{\text{TK_SQL}\}$

and,

$\text{FOLLOW}(\langle \text{optionalReturn} \rangle) = \{\text{TK_SEM}\}$

.....using 42

Since both of the above sets are disjoint, the grammar for the nonterminal $\langle \text{optionalReturn} \rangle$ is LL(1) compatible.

44. $\langle \text{idList} \rangle \Rightarrow \text{TK_ID} \langle \text{more_ids} \rangle$

$\text{FIRST}(\langle \text{idList} \rangle) = \{\text{TK_ID}\}$ and the production is not nullable, hence the rule is LL(1) compatible.

45. $\langle \text{more_ids} \rangle \Rightarrow \text{TK_COMMA} \langle \text{idList} \rangle \mid \text{eps}$

$\text{FIRST}(\text{TK_COMMA} \langle \text{idList} \rangle) = \{\text{TK_COMMA}\}$

and,

$\text{FOLLOW}(\langle \text{more_ids} \rangle) = \text{FOLLOW}(\langle \text{idList} \rangle)$

.....using 44

$$= \{TK_SQR\}$$

.....using 43

This implies that

$$\text{FIRST}(TK_COMMA \langle idList \rangle) \cap \text{FOLLOW}(\langle more_ids \rangle)$$

$$= \{TK_COMMA\} \cap \{TK_SQR\}$$

$$= \phi$$

Therefore, the rules for the nonterminal $\langle more_ids \rangle$ are LL(1) compatible.

Vandana

February 20, 2022